

Sensitivity Analysis Using Different Concepts of Income

Goal

When analysing inequality, you should always utilise multiple methods when comparing different countries. Inequality measures can be affected by the indicator used or by the measure of income.

The Gini index may change enormously when it is calculated on income concepts other than disposable income (*dpi*), which is calculated after transfers and taxes. Instead, you may want to look at market income (*mi*), which is calculated before taxes and transfers. By computing a Gini index on both income concepts, you can analyse the effect of government influence on the income distribution.

Activity

Calculate the Gini index for Finland and the US in 2000 for both market income and total disposable income, after bottom and top coding as in the previous exercise.

Guidelines

- Prepare the data as you did in the previous exercise. Be careful when choosing your sample. For each of the two income concepts, there will be a different sample of “valid” values (i.e., not missing and not zero). Since we are comparing the two final measures, we want to use the same sample in the two cases. In this exercise, use the sample that drops observations for which *dpi* is zero or missing. In this case, observations with zero *mi* but “valid” *dpi* are kept.
- In this exercise, you will need to create two measures of income: an equivalised disposable income (*dpi*) as well as an equivalised market income (*mi*).

Since you are carrying out the same commands for both income measures you could create a **foreach** loop for the two variables (within the other **foreach** loop for the two datasets):

```
foreach var in mi dpi {
    display "Analysis using `var'"
    gen e`var'=(`var'/(d4^0.5))
    qui sum e`var' [w=hweight*d4]
    replace e`var' = .01*r(mean) if e`var'<.01*r(mean)
    qui sum `var' [w=hweight*d4], de
    replace e`var' = (10*r(p50)/(d4^0.5)) if `var'>10*r(p50)
    ineqdeco e`var' [w=hweight*d4]
}
```

Program

```
di "*** INCOME DISTRIBUTION II - Exercise 14 ***"
global varlist "mi dpi"
foreach file in $us00h $fi00h {
  display "`file'"
  use hweight d4 d5 dpi mi if (!mi(dpi) & !(dpi==0)) using `file', clear
  foreach var of global varlist {
    display "Analysis using `var'"
    gen e`var'=(`var'/(d4^0.5))
    qui sum e`var' [w=hweight*d4]
    replace e`var' = .01*r(mean) if e`var'<.01*r(mean)
    qui sum `var' [w=hweight*d4], de
    replace e`var' = (10*r(p50)/(d4^.5)) if `var'>10*r(p50)
    ineqdeco e`var' [w=hweight*d4]
  }
}
```

Results

Gini	<i>mi</i>	<i>dpi</i>
US 2000	0.478	0.368
Finland 2000	0.463	0.246

Comments

- The Gini index is obviously much larger when calculated on the market income than on net income, since both transfers and taxes have a redistributive purpose.
- Even though Finland remains the more equal of the two countries even if looking at the *mi* measure, the difference becomes much less evident; the much larger decrease in Gini (when going from the pre-tax and transfers to the post-tax and transfer income concept) in Finland with respect to the US, points to two completely different social systems in terms of income redistribution.
- As already mentioned, there are quite a few LIS datasets that contain net income instead of gross income data, including some large countries like France and Italy. The income data from these countries are net of taxes and thus not readily comparable to the gross income data from other countries.